<u>Exemplo de integral de fluxo</u>

<u>Problema:</u>
Calcular o fluxo total do campo gravitacional normalizado através de um cilindro de raio a e altura 2H, centrado na origem alinhado com o eixo y .

Defina o campo de vetores

>

> **G:=[-x/(x^2+y^2+z^2)^(3/2),-y/(x^2+y^2+z^2)^(3/2),-z/(x^2+y^2+z^2)^(3/2)];**

$$G := \left[-\frac{x}{(x^2+y^2+z^2)^{3/2}}, -\frac{y}{(x^2+y^2+z^2)^{3/2}}, -\frac{z}{(x^2+y^2+z^2)^{3/2}} \right]$$

Vamos desenha-lo.

> **with(plots):**
fieldplot3d(G,x=-3..3,y=-3..3,z=-3..3,grid=[4,4,4],
axes=normal,thickness=3);



>

Parametrização dos discos extremos

Primeiro parametrizamos o disco a direita. Usamos coordenadas polares.

```
> x1st:=s*cos(t);
y1st:=H;
z1st:=s*sin(t);
R1:=[x1st,y1st,z1st];
srange1:=s=0..a;
trange1:=t=0..2*Pi;
```

$$x1st := s\cos(t)$$

$$y1st := H$$

$$z1st := s\sin(t)$$

$$R1 := [s\cos(t), H, s\sin(t)]$$

$$srange1 := s = 0 .. a$$

$$trange1 := t = 0 .. 2\pi$$

```
> rightend:=plot3d(subs(a=2,H=3,[x1st,y1st,z1st]),
subs(a=2,H=3,srange1),trange1,grid=[4,12],
style=patch,axes=normal,view=[-4..4,-4..4,-4..4]):
rightend;
```



O disco da esquerda é similar, simplesmente substitua H por -H.
Note a orientação oposta sobre este disco.

```
> x2st:=s*cos(t);
y2st:=-H;
z2st:=s*sin(t);
R2:=[x2st,y2st,z2st];
srange2:=s=0..a;
trange2:=t=0..2*Pi;
```

$$x2st := s\cos(t)$$

$$y2st := -H$$

$$z2st := s\sin(t)$$

$$R2 := [s\cos(t), -H, s\sin(t)]$$

$$srange2 := s = 0 .. a$$

$$trange2 := t = 0 .. 2\pi$$

```
> leftend:=plot3d(subs(a=2,H=3,[x2st,y2st,z2st]),
subs(a=2,H=3,srange2),trange2,grid=[4,12],
style=patch,axes=normal,view=[-4..4,-4..4,-4..4]):
```

```
> display([rightend,leftend]);
```



Parametrize o cilindro

```
> x0st:=a*cos(t);
y0st:=s;
z0st:=a*sin(t);
R0:=[x0st,y0st,z0st];
srange0:=s=-H..H;
trange0:=t=0..2*Pi;
```

$$x0st := a\cos(t)$$

$$y0st := s$$

$$z0st := a\sin(t)$$

$$R0 := [a\cos(t), s, a\sin(t)]$$

$$srange0 := s = -H\,..\,H$$

$$trange0 := t = 0\,..\,2\,\pi$$

```
> wall:=plot3d(subs(a=2,H=3,[x0st,y0st,z0st]),
subs(a=2,H=3,srange0),trange0,grid=[12,12],color=red,thickness=1,
style=wireframe,axes=normal,view=[-4..4,-4..4,-4..4]):
```

```
> display([rightend,leftend,wall],tickmarks=[3,3,3]);
```



Fluxo através do extremo direito

Primeiro lembramos a parametrizacao do lado direito:

> **R1;**

$$[s\cos(t), H, s\sin(t)]$$

Calcule as velocidades com respeito a s e t:

> **R1s:=diff(R1,s);**
R1t:=diff(R1,t);

$$R1s := [\cos(t), 0, \sin(t)]$$

$$R1t := [-s\sin(t), 0, s\cos(t)]$$

Calcule as "areas infinitesimais vetoriais":

> **with(linalg):**

Warning, new definition for norm

Warning, new definition for trace

> **dA1:=crossprod(R1s,R1t);**

$$dA1 := [0, -\sin(t)^2 s - \cos(t)^2 s, 0]$$

Ajuste o sinal de dA para corresponder a orientação do extremo como parte do cilindro.

```
> dA1:=map(simplify,dA1);
dA1:=scalarmul(dA1,-1);
```

$$dA1 := [0, -s, 0]$$

$$dA1 := [0, s, 0]$$

Avalie o campo vetorial sobre a superficie parametrizada: ("pullback" de G para o plano s-t)

```
> G1st:=subs(x=x1st,y=y1st,z=z1st,G);
```

$$G1st := \left[-\frac{s\cos(t)}{(s^2\cos(t)^2+H^2+s^2\sin(t)^2)^{3/2}}, -\frac{H}{(s^2\cos(t)^2+H^2+s^2\sin(t)^2)^{3/2}}, -\frac{s\sin(t)}{(s^2\cos(t)^2+H^2+s^2\sin(t)^2)^{3/2}} \right]$$

```
> G1st:=simplify(G1st);
```

$$G1st := \left[-\frac{s\cos(t)}{(H^2+s^2)^{3/2}}, -\frac{H}{(H^2+s^2)^{3/2}}, -\frac{s\sin(t)}{(H^2+s^2)^{3/2}} \right]$$

Calcule o produto interno no integrando e integre.

```
> flux_right_end:=Int(Int(dotprod(G1st,dA1),srange1),trange1);
```

$$flux\_right\_end := \int_0^{2\pi}\int_0^{a} -\frac{H\,s}{(H^2+s^2)^{3/2}}\,ds\,dt$$

```
> flux1:=simplify(int(int(dotprod(G1st,dA1),srange1),trange1));
```

$$flux1 := 2\,\frac{\pi\,(\mathrm{csgn}(H)\,H-\sqrt{H^2+a^2})\,\mathrm{csgn}(H)}{\sqrt{H^2+a^2}}$$

O fator csgn(H) é devido ao fato que o MAPLE não sabe se H é postivo ou negativo

```
> flux1:=subs(csgn(H)=1,flux1);
```

$$flux1 := 2\,\frac{\pi\,(H-\sqrt{H^2+a^2})}{\sqrt{H^2+a^2}}$$

Para aprender mais:

```
> ##?assume
```

```
> ##?csgn
```

Fluxo através do extremo esquerdo

Por razões de simetria o fluxo deveria ser o mesmo para este extemo.

Lembramos a parametrização do lado esquerdo:

```
> R2;
```

$$[s\cos(t), -H, s\sin(t)]$$

Calcule as "velocidades" com respeito a s e t:

```
> R2s:=diff(R2,s);
R2t:=diff(R2,t);
```

$$R2s := [\cos(t), 0, \sin(t)]$$

$$R2t := [-s\sin(t), 0, s\cos(t)]$$

Calcule as "areas infinitesimais vetoriais":

```
> with(linalg):
```

```
> dA2:=crossprod(R2s,R2t);
```

$$dA2 := [0, -\sin(t)^2\,s-\cos(t)^2\,s, 0]$$

```
> dA2:=map(simplify,dA2);
```

$$dA2 := [0, -s, 0]$$

Avalie o campo vetorial sobre a superficie parametrizada: ("pullback" de G para o plano s-t)

```
> G2st:=subs(x=x2st,y=y2st,z=z2st,G);
```

$$G2st := \left[-\frac{s\cos(t)}{(s^2\cos(t)^2+H^2+s^2\sin(t)^2)^{3/2}}, \frac{H}{(s^2\cos(t)^2+H^2+s^2\sin(t)^2)^{3/2}}, -\frac{s\sin(t)}{(s^2\cos(t)^2+H^2+s^2 \text{s}} \right]$$

```
> G2st:=simplify(G2st);
```

$$G2st := \left[-\frac{s\cos(t)}{(H^2+s^2)^{3/2}}, \frac{H}{(H^2+s^2)^{3/2}}, -\frac{s\sin(t)}{(H^2+s^2)^{3/2}} \right]$$

Calcule o produto interno no integrnado e integre.

```
> flux_left_end:=Int(Int(dotprod(G2st,dA2),srange2),trange2);
```

$$flux\_left\_end := \int_0^{2\pi}\int_0^a -\frac{H\,s}{(H^2+s^2)^{3/2}}\,ds\,dt$$

```
> flux2:=simplify(int(int(dotprod(G2st,dA2),srange2),trange2));
```

$$flux2 := 2\,\frac{\pi\,(\mathrm{csgn}(H)\,H-\sqrt{H^2+a^2})\,\mathrm{csgn}(H)}{\sqrt{H^2+a^2}}$$

Como antes o Maple nao sabe se H eh positivo ou negativo.

```
> flux2:=subs(csgn(H)=1,flux2);
```

$$flux2 := 2\,\frac{\pi\,(H-\sqrt{H^2+a^2})}{\sqrt{H^2+a^2}}$$

>

Fluxo através do tubo

Primeiro lembramos a parametrização:

```
> R0;
```

$$[a\cos(t), s, a\sin(t)]$$

Calcule as "velocidades" com respeito a s e t:

```
> R0s:=diff(R0,s);
R0t:=diff(R0,t);
```

$$R0s := [0, 1, 0]$$

$$R0t := [-a\sin(t), 0, a\cos(t)]$$

Calcule as "areas infinitesimais vetoriais":

```
> with(linalg):
```

```
> dA0:=crossprod(R0s,R0t);
```

$$dA0 := [a\cos(t), 0, a\sin(t)]$$

Neste caso o sinal indica orientação "radialmente exterior " do tubo:

```
> dA0:=map(simplify,dA0);
```

$$dA0 := [a\cos(t), 0, a\sin(t)]$$

Avalie o campo vetorial sobre a superficie parametrizada: ("pullback" de G parap plano s-t)

```
> G0st:=subs(x=x0st,y=y0st,z=z0st,G);
```

$$G0st := \left[-\frac{a\cos(t)}{(a^2\cos(t)^2+s^2+a^2\sin(t)^2)^{3/2}}, -\frac{s}{(a^2\cos(t)^2+s^2+a^2\sin(t)^2)^{3/2}}, -\frac{a\sin(t)}{(a^2\cos(t)^2+s^2+a^2\sin(t)^2)^{3/2}}\right]$$

```
> G0st:=simplify(G0st);
```

$$G0st := \left[-\frac{a\cos(t)}{(s^2+a^2)^{3/2}}, -\frac{s}{(s^2+a^2)^{3/2}}, -\frac{a\sin(t)}{(s^2+a^2)^{3/2}}\right]$$

Calcule o produto interno no integrando e integre.

```
> flux_wall:=Int(Int(simplify(dotprod(G0st,dA0)),srange0),trange0);
```

$$flux\_wall := \int_0^{2\pi} \int_{-H}^{H} -\frac{a^2}{(s^2+a^2)^{3/2}}\, ds\, dt$$

> **flux0:=int(int(simplify(dotprod(G0st,dA0)),srange0),trange0);**

$$flux0 := -4\frac{\pi H}{\sqrt{H^2+a^2}}$$

Fluxo total

> **flux:=flux1+flux2+flux0;**

$$flux := 4\frac{\pi\,(H-\sqrt{H^2+a^2})}{\sqrt{H^2+a^2}} - 4\frac{\pi H}{\sqrt{H^2+a^2}}$$

> **flux:=simplify(flux);**

$$flux := -4\pi$$

Calcule o divergente do campo vetorial

O divergente é dado por :

> **divG:=diff(G[1],x)+diff(G[2],y)+diff(G[3],z);**

$$divG := -\frac{3}{(x^2+y^2+z^2)^{3/2}} + 3\frac{x^2}{(x^2+y^2+z^2)^{5/2}} + 3\frac{y^2}{(x^2+y^2+z^2)^{5/2}} + 3\frac{z^2}{(x^2+y^2+z^2)^{5/}}$$

> **divG:=simplify(divG);**

$$divG := 0$$

Note que div(G) não está definido na origem e assim o teorema da divergencia não se aplica imediatamente para o fluxo no problema.

> **with(linalg):**

> **diverge(G,[x,y,z]);**

$$-\frac{3}{(x^2+y^2+z^2)^{3/2}}+3\frac{x^2}{(x^2+y^2+z^2)^{5/2}}+3\frac{y^2}{(x^2+y^2+z^2)^{5/2}}+3\frac{z^2}{(x^2+y^2+z^2)^{5/2}}$$

```
> curl(G,[x,y,z]);
```

$$[0, 0, 0]$$

Usando o teorema da divergência

O divergente div(G) e o campo G não estão definidos na origem.
Mas em todo o lugar a divergência é zero.

Consequentemente a integral (triple) do divergente de G sobre qualquer região do espaço que não inclui a origem é zero.

Portanto o fluxo total através da superficie inteira (orientada positivamente) que limita qualquer região no espaço que não contém a origem também é zero.

A estratégia é encontrar a região no espaço que é limitada pela superficie para a qual queremos calcular o fluxo, mais

alguma outra superficie para a qual o luxo é trivial para calcular.

A região no espaço está na regiao entre a esfera e o cilindro (ambos superficies e não objetos solidos)

```
> unitsphere:=plot3d([cos(t)*sin(s),sin(t)*sin(s),cos(s)],t=0..2*Pi,s=0..Pi,
style=patch,color=blue,grid=[12,6]):
display([unitsphere,rightend,leftend,wall],tickmarks=[3,3,3]);
```



O fluxo sobre a esfera eh igual a -4π : o campo vetorial G tem magnitude constante sobre a esfera e é sempre perpendicular a superficie. Assim o fluxo total é igual a area da esfera menos o sinal para corrigir a direção de G (radialmente interior) vezes a magnetude (constante) de G sobre a esfera que é igual a 1.

```
> x00st:=cos(t)*sin(s);
y00st:=sin(t)*sin(s);
z00st:=cos(s);
R00:=[x00st,y00st,z00st];
srange00:=s=0..Pi;
trange00:=t=0..2*Pi;
```

$$x00st := \cos(t)\sin(s)$$

$$y00st := \sin(t)\sin(s)$$

$$z00st := \cos(s)$$

$$R00 := [\cos(t)\sin(s), \sin(t)\sin(s), \cos(s)]$$

$$srange00 := s = 0 \,..\, \pi$$

$$trange00 := t = 0 \,..\, 2\pi$$

Calcule os vetores velocidades:

```
> R00s:=diff(R00,s);
R00t:=diff(R00,t);
```

$$R00s := [\cos(t)\cos(s), \sin(t)\cos(s), -\sin(s)]$$

$$R00t := [-\sin(t)\sin(s), \cos(t)\sin(s), 0]$$

Calcule as "areas infinitesimais vetoriais":

```
> with(linalg):
```

```
> dA00:=crossprod(R00s,R00t);
```

$$dA00 := [\sin(s)^2\cos(t), \sin(s)^2\sin(t), \cos(t)^2\cos(s)\sin(s) + \sin(t)^2\cos(s)\sin(s)]$$

```
> dA00:=map(simplify,dA00);
```

$$dA00 := [\cos(t) - \cos(t)\cos(s)^2, \sin(t) - \sin(t)\cos(s)^2, \cos(s)\sin(s)]$$

Evalie o campo vetorial sobre a superficie parametrizada ("pullback" de G para o plano s-t)

```
> G00st:=subs(x=x00st,y=y00st,z=z00st,G);
```

$$G00st := \left[-\frac{\cos(t)\sin(s)}{(\cos(t)^2\sin(s)^2 + \sin(t)^2\sin(s)^2 + \cos(s)^2)^{3/2}}, -\frac{\sin(t)\sin(s)}{(\cos(t)^2\sin(s)^2 + \sin(t)^2\sin(s)^2 + \cos(s)^2)^{3/2}}, -\frac{\cos(s)}{(\cos(t)^2\sin(s)^2 + \sin(t)^2\sin(s)^2 + \cos(s)^2)^{3/2}}\right]$$

```
> G00st:=simplify(G00st);
```

$$G00st := [-\cos(t)\sin(s), -\sin(t)\sin(s), -\cos(s)]$$

Calcule o produto interno no integrndo e integre.

```
> flux_sphere:=Int(Int(simplify(dotprod(G00st,dA00)),srange00),trange00);
```

$$flux\_sphere := \int_0^{2\pi}\int_0^{\pi} -\sin(s)\, ds\, dt$$

```
> fluxsphere:=int(int(simplify(dotprod(G00st,dA00)),srange00),trange00);
```

$$fluxsphere := -4\pi$$

```
>
```